shootface_Netlogo_Minecraft/Models/BicicletasUD.nlogo
globals [
  attendance        ;; el aforo actual al ciclo-parqueadero
  history           ;; lista de valores historicos de aforo
  home-patches      ;; patches verdes (hogares de los bici-usuarios)
  bicycle-parking-patches      ;; patches azules (ciclo-parqueadero)
  crowded-patch     ;; patch de etiqueta de sobrecupo en el ciclo-parqueadero
  ud-pacth-label    ;; patch de etiqueta de la UD
  home-patch-label  ;; patch de etiqueta de los hogares
]

turtles-own [
  strategies      ;; lista de estrategias
  best-strategy   ;; indice de la mejor estrategia actual
  attend?         ;; es true si el agente planea ir al ciclo-parqueadero en ese momento
  prediction      ;; prediccion actual del aforo del ciclo-parqueadero
]

to setup
  clear-all
  set-default-shape turtles "bike"
  ;; crear patches verdes (hogares de los bici-usuarios)
  set home-patches patches with [pycor < 0 or (pxcor <  0 and pycor >= 0)]
  ask home-patches [ set pcolor green ]

  ;; crear patches azules (ciclo-parqueadero)
  set bicycle-parking-patches patches with [pxcor > 0 and pycor > 0]
  ask bicycle-parking-patches [ set pcolor blue ]

  ;; crear etiquetas de UD y hogares
  ask patch 21 2 [
    set plabel "Ciclo-parqueadero"
    set plabel-color black
  ]

  ask patch 24 24 [
    set plabel "Universidad Distrital FJDC"
    set plabel-color black
  ]

  ask patch 11 -24 [
    set plabel "Hogares bici-usuarios"
    set plabel-color black
  ]

  ;; inicializar la el aforo al azar para que los agentes tengan un historial con el que hacer predicciones desde el principio
  set history n-values (memoria * 2) [random 100]
  ;; el historico debe ser el doble de la memoria

  set attendance first history

  ;; etiqueta de los patches del ciclo-parqueadero para indicar visualmente si esta lleno o no
  ask patch (0.75 * max-pxcor) (0.5 * max-pycor) [
    set crowded-patch self
    set plabel-color red
  ]

  ;; crear los agentes y darles estrategias aleatorias
  create-turtles 100 [
    set color white
    set size 2
    move-to-empty-one-of home-patches
    set strategies n-values cantidad-estrategias [random-strategy]
    set best-strategy first strategies
    update-strategies
  ]

  reset-ticks
  exportFile
end


to go
  ask crowded-patch [ set plabel "" ]
  ;; cada agente predice la el aforo del ciclo-parqueadero y decide si ir o buscar otro lugar primero
  ask turtles [
    set prediction predict-attendance best-strategy sublist history 0 memoria
    set attend? (prediction <= umbral-sobrecupo)  ;; true or false
  ]
  ;; dependiendo de su decision, el agente se movera
  ask turtles [
    ifelse attend?
      [ move-to-empty-one-of bicycle-parking-patches
        set attendance attendance + 1 ]
      [ move-to-empty-one-of home-patches ]
  ]

  ;; si hay sobrecupo, activar la etiqueta
  set attendance count turtles-on bicycle-parking-patches
  if attendance > umbral-sobrecupo [
    ask crowded-patch [ set plabel "SobreCupo" ]
  ]
  ;; actualizar el historial de aforo, eliminar el aforo mas antiguo y poner el aforo mas reciente en el historico
  set history fput attendance but-last history
  ;; los agentes deciden cual es la nueva mejor estrategia
  ask turtles [ update-strategies ]
  tick
end

;; determinar que estrategia habria predicho los mejores resultados si se hubiera utilizado en este tick
to update-strategies
  ;; inicializar el mejor puntaje al maximo, que es el puntaje mas bajo posible
  let best-score memoria * 100 + 1
  foreach strategies [ the-strategy ->
    let score 0
    let week 1
    repeat memoria [
      set prediction predict-attendance the-strategy sublist history week (week + memoria)
      set score score + abs (item (week - 1) history - prediction)
      set week week + 1
    ]
    if (score <= best-score) [
      set best-score score
      set best-strategy the-strategy
    ]
  ]
end

;; estrategia aleatoria, una estrategia es solo un conjunto de pesos de -1.0 a 1.0
to-report random-strategy
  report n-values (memoria + 1) [1.0 - random-float 2.0]
end

;; prediccion de un agente del aforo actual utilizando una estrategia particular y una parte del historial de aforo
to-report predict-attendance [strategy subhistory]
  report 100 * first strategy + sum (map [ [weight week] -> weight * week ] butfirst strategy subhistory)
end

;; no importa exactamente en que patch esta un agente, solo esta en azules o en verdes.
;; para tener una buena visualización, este procedimiento se utiliza para garantizar que solo se tenga un agente por patch
to move-to-empty-one-of [locations]
  move-to one-of locations
  while [any? other turtles-here] [
    move-to one-of locations
  ]
end

to exportFile
  export-world "Data/word.json"
end
@#$#@#$#@
GRAPHICS-WINDOW
298
10
918
631
-1
-1
12.0
1
24
1
1
1
0
1
1
1
-25
25
-25
25
1
1
1
ticks
30.0

BUTTON
155
208
235
253
Go
go
T
1
T
OBSERVER
NIL
NIL
NIL
NIL
0

BUTTON
56
208
141
254
Set Up
setup
NIL
1
T
OBSERVER
NIL
NIL
NIL
NIL
1

SLIDER
60
40
250
73
memoria
memoria
1
10
5.0
1
1
NIL
HORIZONTAL

SLIDER
60
75
250
108
cantidad-estrategias
cantidad-estrategias
1
20
10.0
1
1
NIL
HORIZONTAL

PLOT
15
345
289
557
Aforo del Ciclo-parqueadero
Tiempo
Aforo
0.0
10.0
0.0
100.0
true
false
"" ""
PENS
"attendance" 1.0 0 -16777216 true "" "plot attendance"
"threshold" 1.0 0 -2674135 true "" ";; plot a threshold line -- an attendance level above this line makes the bar\n;; is unappealing, but below this line is appealing\nplot-pen-reset\nplotxy 0 umbral-sobrecupo\nplotxy plot-x-max umbral-sobrecupo"

SLIDER
60
110
250
143
umbral-sobrecupo
umbral-sobrecupo
0
100
60.0
1
1
NIL
HORIZONTAL

@#$#@#$#@
# Ciclo-parqueadero y sobrecupo (Universidad Distrital - Facultad de Ingeniería)
## INTRODUCCIÓN

Este modelo está basado en las simulaciones propuestas por Uri Wilensky y William Rand en el libro  "Introduction to Agent-Based Modeling: Modeling Natural, Social and Engineered Complex Systems with NetLogo".

## ¿QUÉ PROBLEMA DESCRIBE EL MODELO?

La Facultad de Ingeniería de la Universidad Distrital FJDC cuenta con un pequeño parqueadero para los bici-usuarios, ubicado cerca de la entrada de la carrera octava. Debido al hacinamiento y a la gran cantidad de estudiantes y profesores que se movilizan desde y hacia la universidad en bicicleta, este pequeño espacio no es lo suficientemente grande, ni cuenta con los protocolos de vigilancia necesarios para dejar allí sus vehículos de forma organizada y segura.

El ciclo-parqueadero es usado por gran cantidad de personas en el trascurso de los 6 días de la semana, en los que hay actividades académicas y administrativas. Además, hay horas en específico en que se presenta sobrecupo y no es fácil guardar allí las bicicletas. De hecho, si los bici-usuarios concluyen que no hay posibilidad de guardar sus vehículos, trataran de buscar otros lugares cercanos donde dejarlos (amarrados en árboles, postes, rejas u otros parqueaderos en donde hay que pagar). De lo contrario dejan sus bicicletas en el espacio asignado bajo la "vigilancia" del celador encargado. 

Este modelo explora lo que sucede con el uso general del ciclo-parqueadero en las horas de mas congestión y movimiento de estudiantes y maestros a lo largo de la semana, ya que usualmente utilizan diferentes estrategias para determinar qué tan lleno creen que estará el espacio cuando salen de sus hogares o trabajos para dirigirse a la universidad.

## ¿CÓMO FUNCIONA LA SIMULACIÓN?

Un agente irá directamente al lugar del ciclo-parqueadero si cree que no habrá más de un cierto número de bicicletas allí, este es el umbral de sobrecupo. Para predecir la congestión de bicicletas en el parqueadero para cada día de la semana, cada agente tiene acceso a un conjunto de estrategias de predicción y a las cifras de aforo del ciclo-parqueadero del día en curso en semanas anteriores. Una estrategia de predicción se representa como una lista de pesos que determina cómo el agente cree que cada período de tiempo en los datos históricos afecta la predicción de asistencia para el día actual. El agente decide cuál de sus estrategias utilizar mediante la determinación de cuál habría sido mejor si la hubiera utilizado en las semanas anteriores.

Curiosamente, la estrategia óptima desde un punto de vista perfectamente racional sería ir siempre al ciclo-parqueadero, ya que no se pierde mucho por ir cuando está lleno, pero en este modelo los agentes tratan de predecir cuál será el aforo para ese día y si es viable ir al ciclo-parqueadero o buscar otro lugar primero.

El número de estrategias potenciales que tiene un agente se distribuyen aleatoriamente a los agentes durante el proceso de configuración inicial. A medida que se ejecuta el modelo, cada agente solo utilizará una estrategia, en función de su capacidad previa para predecir el aforo del ciclo-parqueadero. Dado que los agentes pueden cambiar sus estrategias en cualquier momento en función del rendimiento, el conjunto de las estrategias utilizadas por toda la población cambia con el tiempo.

## ¿CÓME SE USA LA SIMULACIÓN?

La variable CANTIDAD-ESTRATEGIAS controla cuántas estrategias mantiene cada agente en su memoria. La variable UMBRAL-SOBRECUPO controla cuando el ciclo-parqueadero se considera inviable para guardar la bicicleta. La variable MEMORIA controla que tan atrás en el tiempo pueden los agentes tener registros históricos para hacer sus predicciones. 

Para ejecutar el modelo, se establece la CANTIDAD-ESTRATEGIAS, el UMBRAL-SOBRECUPO y el tamaño de la MEMORIA, por último se presiona SETUP, y luego GO. El diagrama de AFORO-CICLOPARQUEADERO muestra el promedio del aforo al ciclo-parqueadero a lo largo del tiempo.


## ENTORNO O MUNDO DEL PROBLEMA

La parte verde del mundo representa los hogares de los bici-usuarios, mientras que la parte azul del mundo representa el ciclo-parqueadero. Con el tiempo, el aforo aumentará y disminuirá, pero su valor medio se acerca al UMBRAL-SOBRECUPO.

## CONFIGURACIONES INICIALES

Intentar ejecutar el modelo con diferentes configuraciones para MEMORIA y CANTIDAD-ESTRATEGIAS. ¿Qué sucede con la variabilidad en el aforo a medida que disminuye CANTIDAD-ESTRATEGIAS? ¿Qué sucede con la variabilidad en el gráfico si disminuye MEMORIA?

## AUTORES

Marlon Arias Cárdenas - 20151020111
Juan Camilo Guaba - 20151020047

## BIBLIOGRAFÍA

Wilensky, U. & Rand, W. (2015). Introduction to Agent-Based Modeling: Modeling Natural, Social and Engineered Complex Systems with NetLogo. Cambridge, MA. MIT Press.
@#$#@#$#@
default
true
0
Polygon -7500403 true true 150 5 40 250 150 205 260 250

airplane
true
0
Polygon -7500403 true true 150 0 135 15 120 60 120 105 15 165 15 195 120 180 135 240 105 270 120 285 150 270 180 285 210 270 165 240 180 180 285 195 285 165 180 105 180 60 165 15

arrow
true
0
Polygon -7500403 true true 150 0 0 150 105 150 105 293 195 293 195 150 300 150

bike
false
1
Line -7500403 false 163 183 228 184
Circle -7500403 false false 213 184 22
Circle -7500403 false false 156 187 16
Circle -16777216 false false 28 148 95
Circle -16777216 false false 24 144 102
Circle -16777216 false false 174 144 102
Circle -16777216 false false 177 148 95
Polygon -2674135 true true 75 195 90 90 98 92 97 107 192 122 207 83 215 85 202 123 211 133 225 195 165 195 164 188 214 188 202 133 94 116 82 195
Polygon -2674135 true true 208 83 164 193 171 196 217 85
Polygon -2674135 true true 165 188 91 120 90 131 164 196
Line -7500403 false 159 173 170 219
Line -7500403 false 155 172 166 172
Line -7500403 false 166 219 177 219
Polygon -16777216 true false 187 92 198 92 208 97 217 100 231 93 231 84 216 82 201 83 184 85
Polygon -7500403 true true 71 86 98 93 101 85 74 81
Rectangle -16777216 true false 75 75 75 90
Polygon -16777216 true false 70 87 70 72 78 71 78 89
Circle -7500403 false false 153 184 22
Line -7500403 false 159 206 228 205

box
false
0
Polygon -7500403 true true 150 285 285 225 285 75 150 135
Polygon -7500403 true true 150 135 15 75 150 15 285 75
Polygon -7500403 true true 15 75 15 225 150 285 150 135
Line -16777216 false 150 285 150 135
Line -16777216 false 150 135 15 75
Line -16777216 false 150 135 285 75

bug
true
0
Circle -7500403 true true 96 182 108
Circle -7500403 true true 110 127 80
Circle -7500403 true true 110 75 80
Line -7500403 true 150 100 80 30
Line -7500403 true 150 100 220 30

butterfly
true
0
Polygon -7500403 true true 150 165 209 199 225 225 225 255 195 270 165 255 150 240
Polygon -7500403 true true 150 165 89 198 75 225 75 255 105 270 135 255 150 240
Polygon -7500403 true true 139 148 100 105 55 90 25 90 10 105 10 135 25 180 40 195 85 194 139 163
Polygon -7500403 true true 162 150 200 105 245 90 275 90 290 105 290 135 275 180 260 195 215 195 162 165
Polygon -16777216 true false 150 255 135 225 120 150 135 120 150 105 165 120 180 150 165 225
Circle -16777216 true false 135 90 30
Line -16777216 false 150 105 195 60
Line -16777216 false 150 105 105 60

car
false
0
Polygon -7500403 true true 300 180 279 164 261 144 240 135 226 132 213 106 203 84 185 63 159 50 135 50 75 60 0 150 0 165 0 225 300 225 300 180
Circle -16777216 true false 180 180 90
Circle -16777216 true false 30 180 90
Polygon -16777216 true false 162 80 132 78 134 135 209 135 194 105 189 96 180 89
Circle -7500403 true true 47 195 58
Circle -7500403 true true 195 195 58

circle
false
0
Circle -7500403 true true 0 0 300

circle 2
false
0
Circle -7500403 true true 0 0 300
Circle -16777216 true false 30 30 240

cow
false
0
Polygon -7500403 true true 200 193 197 249 179 249 177 196 166 187 140 189 93 191 78 179 72 211 49 209 48 181 37 149 25 120 25 89 45 72 103 84 179 75 198 76 252 64 272 81 293 103 285 121 255 121 242 118 224 167
Polygon -7500403 true true 73 210 86 251 62 249 48 208
Polygon -7500403 true true 25 114 16 195 9 204 23 213 25 200 39 123

cylinder
false
0
Circle -7500403 true true 0 0 300

dot
false
0
Circle -7500403 true true 90 90 120

face happy
false
0
Circle -7500403 true true 8 8 285
Circle -16777216 true false 60 75 60
Circle -16777216 true false 180 75 60
Polygon -16777216 true false 150 255 90 239 62 213 47 191 67 179 90 203 109 218 150 225 192 218 210 203 227 181 251 194 236 217 212 240

face neutral
false
0
Circle -7500403 true true 8 7 285
Circle -16777216 true false 60 75 60
Circle -16777216 true false 180 75 60
Rectangle -16777216 true false 60 195 240 225

face sad
false
0
Circle -7500403 true true 8 8 285
Circle -16777216 true false 60 75 60
Circle -16777216 true false 180 75 60
Polygon -16777216 true false 150 168 90 184 62 210 47 232 67 244 90 220 109 205 150 198 192 205 210 220 227 242 251 229 236 206 212 183

fish
false
0
Polygon -1 true false 44 131 21 87 15 86 0 120 15 150 0 180 13 214 20 212 45 166
Polygon -1 true false 135 195 119 235 95 218 76 210 46 204 60 165
Polygon -1 true false 75 45 83 77 71 103 86 114 166 78 135 60
Polygon -7500403 true true 30 136 151 77 226 81 280 119 292 146 292 160 287 170 270 195 195 210 151 212 30 166
Circle -16777216 true false 215 106 30

flag
false
0
Rectangle -7500403 true true 60 15 75 300
Polygon -7500403 true true 90 150 270 90 90 30
Line -7500403 true 75 135 90 135
Line -7500403 true 75 45 90 45

flower
false
0
Polygon -10899396 true false 135 120 165 165 180 210 180 240 150 300 165 300 195 240 195 195 165 135
Circle -7500403 true true 85 132 38
Circle -7500403 true true 130 147 38
Circle -7500403 true true 192 85 38
Circle -7500403 true true 85 40 38
Circle -7500403 true true 177 40 38
Circle -7500403 true true 177 132 38
Circle -7500403 true true 70 85 38
Circle -7500403 true true 130 25 38
Circle -7500403 true true 96 51 108
Circle -16777216 true false 113 68 74
Polygon -10899396 true false 189 233 219 188 249 173 279 188 234 218
Polygon -10899396 true false 180 255 150 210 105 210 75 240 135 240

house
false
0
Rectangle -7500403 true true 45 120 255 285
Rectangle -16777216 true false 120 210 180 285
Polygon -7500403 true true 15 120 150 15 285 120
Line -16777216 false 30 120 270 120

leaf
false
0
Polygon -7500403 true true 150 210 135 195 120 210 60 210 30 195 60 180 60 165 15 135 30 120 15 105 40 104 45 90 60 90 90 105 105 120 120 120 105 60 120 60 135 30 150 15 165 30 180 60 195 60 180 120 195 120 210 105 240 90 255 90 263 104 285 105 270 120 285 135 240 165 240 180 270 195 240 210 180 210 165 195
Polygon -7500403 true true 135 195 135 240 120 255 105 255 105 285 135 285 165 240 165 195

line
true
0
Line -7500403 true 150 0 150 300

line half
true
0
Line -7500403 true 150 0 150 150

pentagon
false
0
Polygon -7500403 true true 150 15 15 120 60 285 240 285 285 120

person
false
0
Circle -7500403 true true 110 5 80
Polygon -7500403 true true 105 90 120 195 90 285 105 300 135 300 150 225 165 300 195 300 210 285 180 195 195 90
Rectangle -7500403 true true 127 79 172 94
Polygon -7500403 true true 195 90 240 150 225 180 165 105
Polygon -7500403 true true 105 90 60 150 75 180 135 105

plant
false
0
Rectangle -7500403 true true 135 90 165 300
Polygon -7500403 true true 135 255 90 210 45 195 75 255 135 285
Polygon -7500403 true true 165 255 210 210 255 195 225 255 165 285
Polygon -7500403 true true 135 180 90 135 45 120 75 180 135 210
Polygon -7500403 true true 165 180 165 210 225 180 255 120 210 135
Polygon -7500403 true true 135 105 90 60 45 45 75 105 135 135
Polygon -7500403 true true 165 105 165 135 225 105 255 45 210 60
Polygon -7500403 true true 135 90 120 45 150 15 180 45 165 90

square
false
0
Rectangle -7500403 true true 30 30 270 270

square 2
false
0
Rectangle -7500403 true true 30 30 270 270
Rectangle -16777216 true false 60 60 240 240

star
false
0
Polygon -7500403 true true 151 1 185 108 298 108 207 175 242 282 151 216 59 282 94 175 3 108 116 108

target
false
0
Circle -7500403 true true 0 0 300
Circle -16777216 true false 30 30 240
Circle -7500403 true true 60 60 180
Circle -16777216 true false 90 90 120
Circle -7500403 true true 120 120 60

tree
false
0
Circle -7500403 true true 118 3 94
Rectangle -6459832 true false 120 195 180 300
Circle -7500403 true true 65 21 108
Circle -7500403 true true 116 41 127
Circle -7500403 true true 45 90 120
Circle -7500403 true true 104 74 152

triangle
false
0
Polygon -7500403 true true 150 30 15 255 285 255

triangle 2
false
0
Polygon -7500403 true true 150 30 15 255 285 255
Polygon -16777216 true false 151 99 225 223 75 224

truck
false
0
Rectangle -7500403 true true 4 45 195 187
Polygon -7500403 true true 296 193 296 150 259 134 244 104 208 104 207 194
Rectangle -1 true false 195 60 195 105
Polygon -16777216 true false 238 112 252 141 219 141 218 112
Circle -16777216 true false 234 174 42
Rectangle -7500403 true true 181 185 214 194
Circle -16777216 true false 144 174 42
Circle -16777216 true false 24 174 42
Circle -7500403 false true 24 174 42
Circle -7500403 false true 144 174 42
Circle -7500403 false true 234 174 42

turtle
true
0
Polygon -10899396 true false 215 204 240 233 246 254 228 266 215 252 193 210
Polygon -10899396 true false 195 90 225 75 245 75 260 89 269 108 261 124 240 105 225 105 210 105
Polygon -10899396 true false 105 90 75 75 55 75 40 89 31 108 39 124 60 105 75 105 90 105
Polygon -10899396 true false 132 85 134 64 107 51 108 17 150 2 192 18 192 52 169 65 172 87
Polygon -10899396 true false 85 204 60 233 54 254 72 266 85 252 107 210
Polygon -7500403 true true 119 75 179 75 209 101 224 135 220 225 175 261 128 261 81 224 74 135 88 99

wheel
false
0
Circle -7500403 true true 3 3 294
Circle -16777216 true false 30 30 240
Line -7500403 true 150 285 150 15
Line -7500403 true 15 150 285 150
Circle -7500403 true true 120 120 60
Line -7500403 true 216 40 79 269
Line -7500403 true 40 84 269 221
Line -7500403 true 40 216 269 79
Line -7500403 true 84 40 221 269

x
false
0
Polygon -7500403 true true 270 75 225 30 30 225 75 270
Polygon -7500403 true true 30 75 75 30 270 225 225 270
@#$#@#$#@
NetLogo 6.1.1
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
default
0.0
-0.2 0 0.0 1.0
0.0 1 1.0 0.0
0.2 0 0.0 1.0
link direction
true
0
Line -7500403 true 150 150 90 180
Line -7500403 true 150 150 210 180
@#$#@#$#@
1
@#$#@#$#@
